# $Id: PKGBUILD 90113 2013-05-06 19:40:18Z foutrelis $
# Maintainer: Sergej Pupykin <pupykin.s+arch@gmail.com>
# Maintainer: Eric Belanger <belanger@astro.umontreal.ca>
# Contributor: Link Dupont <link@subpop.net>

pkgname=xloadimage
pkgver=4.1
pkgrel=13
pkgdesc="An utility to view many different types of images under X11"
arch=(i686 x86_64)
#url="http://world.std.com/~jimf/xloadimage.html"
url="http://sioseis.ucsd.edu/xloadimage.html"
license=("MIT")
depends=('libtiff' 'libpng' 'libx11' 'libxext')
makedepends=('patch' 'imake')
source=(http://archlinux-stuff.googlecode.com/files/xloadimage.$pkgver-1.tar.gz
	license.txt
	enable-image-types.patch
	png15-tiff4.patch)
md5sums=('86b42b1b628a9c00008b7deec21e1175'
         'cc16a51aeb5083b8ff07939dfe244130'
         '3a2703925011276fd2d2bd139b676487'
         'c797cfc3f4588b96275d82c2c9cc7025')

build() {
  cd $srcdir/$pkgname.$pkgver

  patch -p1 <$srcdir/enable-image-types.patch
  patch -p1 <$srcdir/png15-tiff4.patch
  xmkmf
  make
}

package() {
  cd $srcdir/$pkgname.$pkgver

  install -D -m755 xloadimage $pkgdir/usr/bin/xloadimage
  install -D -m755 uufilter $pkgdir/usr/bin/uufilter
  install -D -m644 xloadimagerc $pkgdir/etc/xloadimagerc.example
  install -D -m644 xloadimage.man $pkgdir/usr/man/man1/xloadimage.1x
  install -D -m644 uufilter.man $pkgdir/usr/man/man1/uufilter.1x
  install -D -m644 $srcdir/license.txt $pkgdir/usr/share/licenses/$pkgname/license.txt
  ln -s /usr/bin/xloadimage $pkgdir/usr/bin/xsetbg

  mv $pkgdir/usr/man $pkgdir/usr/share/
}
